

## A Validation Macro to Check Compliance of CDISC SDTM Data Hany Aboutaleb, Biogen Idec, Cambridge, MA

### Abstract:

The CDISC SDTM Implementation Guide provides specifications for 30 domains or more and new domains are always being developed. It is important to check the CDISC website for the latest updates before you begin a new project. This validation macro will check the compliance of the SDTM domains as per the most recent SDTM Implementation Guide (SDTM IG) plus our company specific domains. This check will reduce risk of delays in the submission review process. The checks include: domain name, domain description, variable name, variable order, variable label, variable type, and variable length.

### Background:

Clinical databases require some processing/ mapping prior to being 100% SDTM compliant submission ready files. A validation plan should be in place to ensure that the data converted into SDTM can support the submission. Getting the data into SDTM compliant files will save money and time. It is very important to develop a tool to validate SDTM data contents and attributes for all projects in order to foster standards and ensure high quality in mapping clinical data to CDISC SDTM requirements.

A master SDTM excel file can be generated using the document in SDTM version 3.1.1 or 3.1.2 at [www.cdisc.org](http://www.cdisc.org). The sheet named v311 displays Sequence For Order, Observation Class, Domain Prefix, Variable Name (minus domain prefix), Variable Name, Variable Label, Type, Controlled Terms or Format, Origin, Role CDISC Notes, Core, Reference (See Table 1 below).

Table 1 Master SDTM


|     | Seq. For Order | Observation Class | Domain Prefix | Variable Name (minus domain prefix) | Variable Name | Variable Label                      | Type | Controlled Terms or Format | Origin          | Role              | CDISC Notes                                                                                                                                                                                                                   |
|-----|----------------|-------------------|---------------|-------------------------------------|---------------|-------------------------------------|------|----------------------------|-----------------|-------------------|-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| 1   | 1              | Events            | AE            | STUDYID                             | STUDYID       | Study Identifier                    | Char |                            | CRF             | Identifier        | Unique identifier for a study within the submission.                                                                                                                                                                          |
| 334 | 2              | Events            | AE            | DOMAIN                              | DOMAIN        | Domain Abbreviation                 | Char | **AE                       | Derived         | Identifier        | Two-character abbreviation for the domain most relevant to the observation.                                                                                                                                                   |
| 335 | 3              | Events            | AE            | USUBJID                             | USUBJID       | Unique Subject Identifier           | Char |                            | Sponsor Defined | Identifier        | Unique subject identifier within the submission.                                                                                                                                                                              |
| 336 | 4              | Events            | AE            | SEQ                                 | AESEQ         | Sequence Number                     | Num  |                            | CRF or Derived  | Identifier        | Sequence number given to ensure uniqueness within a dataset for a subject. Can be used to join related records.                                                                                                               |
| 337 | 5              | Events            | AE            | GRPID                               | AEGRPID       | Group ID                            | Char |                            | Sponsor Defined | Identifier        | Used to tie together a block of related records in a single domain to support relationships within the domain and between domains.                                                                                            |
| 338 | 6              | Events            | AE            | REFID                               | AEREPID       | Reference ID                        | Char |                            | Sponsor Defined | Identifier        | Optional internal or external identifier such as a serial number on an SAE reporting form.                                                                                                                                    |
| 339 | 7              | Events            | AE            | SPID                                | AESPID        | Sponsor-Defined Identifier          | Char |                            | Sponsor Defined | Identifier        | Optional Sponsor-defined reference number. Perhaps pre-printed on the CRF as an explicit line identifier or defined in the sponsor's operational database. Example: Line number on a Adverse Events page.                     |
| 340 | 8              | Events            | AE            | TERM                                | AETERM        | Reported Term for the Adverse Event | Char |                            | CRF             | Topic             | Verbatim name of the event.                                                                                                                                                                                                   |
| 341 | 9              | Events            | AE            | MODIFY                              | AEMODIFY      | Modified Reported Term              | Char |                            | Sponsor Defined | Synonym Qualifier | If AETERM is modified, then AEMODIFY will contain the modified text.                                                                                                                                                          |
| 342 | 10             | Events            | AE            | DECCD                               | AEDECCD       | Dictionary-Derived Term             | Char | **                         | Derived         | Synonym Qualifier | Dictionary-derived text description of AETERM or AEMODIFY. Equivalent to the Preferred Term (PT in MedDRA). The sponsor should specify the dictionary name and version in the Sponsor Comments column of the Define document. |
| 343 |                |                   | AE            | CAT                                 | AECAT         | Category for Adverse Event          | Char |                            | Sponsor Defined | Qualifier         | Used to differentiate between related domains.                                                                                                                                                                                |

## UNDERSTANDING the Master SDTM Sheet Domains:

The Master SDTM sheet domains are grouped into classes and domains. Each domain is named with two characters. The domains are contained in these classes: Special Purpose, Interventions, Events, Findings and Trial Design Models. Each domain can have its own related file named **SUPPxx** where the **xx** is the two letter domain abbreviation where the “related” data resides. Each SDTM domain is comprised of variables with specific attributes. Each variable may be: REQUIRED, EXPECTED or PERMISSIBLE. The only key difference between REQUIRED and EXPECTED variables is that EXPECTED variable values are allowed to be missing, while REQUIRED must have a value. PERMISSIBLE variables should ONLY be included when data is collected for that study. SDTM Version 3.1.2 is designed with fewer missing values allowed within each domain compared to previous versions of the SDTM standards.

## ELECTRONIC SUBMISSION PLANNING

When companies submit New Drug Applications (NDAs) to the FDA, it is now recommended that they the data in SDTM format. This recommendation is planned to become a requirement at some point in the future. A validation plan should be in place to ensure that the data converted into SDTM can support the submission. In house tools to validate the submission SDTM can save time and money.



Submission Data/ Output Relationships

### %iSDTM Macro:

- %iSDTM is a SAS macro program that uses CDISC SDTM excel sheets as a valuable tool for validating SDTM domain data sets.
- %iSDTM presents basic strategies and practical methods for validating SDTM domain data sets from clinical data management (CDM) system files. This validation will take less time and use fewer resources to produce acceptable SDTM data sets.
- %iSDTM will produce the validation report in MS Excel XLS format.

### %iSDTM Capabilities for Validating a CDISC SDTM SAS Data Set:

The macro performs the following checks and generates validation report:

- Verifies that all required variables are present in the dataset.
- Reports any variables in the dataset that are not defined in the CDISC domain.
- Reports any variables in the SAS data set that are not defined in the domain.
- Verifies that all domain variables are of the expected data type and proper length, and order, and labels as in SDTM v3.1.1 or v3.1.2.
- Verifies that all domain variables labels of length 40 and data sets labels.

**Cont. %iSDTM Capabilities for Validating a CDISC SDTM SAS Data Set:**

- Optional for: verifies that all required variable fields do not contain missing values. Optional for: detects occurrences of expected variable fields that contain missing values.
- Optional for: detects the conformance of all ISO 8601 assigned values, including date, time, datetime, duration, and interval types.
- Validate SUPPQUAL data sets automatically by validating SDTM+SUPP data (merge with the domain data set), and match SUPPQUAL rows with domain data sets.

**Macro Input parameters:**

|           |                                                                                                                                  |
|-----------|----------------------------------------------------------------------------------------------------------------------------------|
| inlib:    | Name of the libname that has the data set to be processed                                                                        |
| excelout: | Name of the Excel file output (default=isdtm.xls)                                                                                |
| sdm:      | Which version of SDTM the macro will use (default=v311)                                                                          |
| chk_req:  | Verifies that all required variable fields do not contain missing values (default=Y)                                             |
| chk_exp:  | Detect occurrences of expected variable fields that contain missing values (default=Y)                                           |
| chk_iso:  | Detect the conformance of all ISO 8601 assigned values, including date, time, datetime, duration, and interval types (default=Y) |
| Debug:    | Debug the macro (YES/ NO) (default: NO)                                                                                          |
| version:  | Version control of the macro for future use in case of a new macro release (default=1)                                           |

**Sample call:**

```
%isdtm(inlib=TEST,sdm=V311,excelout=drug123);
```

**Sample Output:**

**1. Sheet SDTM Summary Checks:**

|    | A                                                           | B                                     | C                   | D                             | E                           | F                               | G                  |
|----|-------------------------------------------------------------|---------------------------------------|---------------------|-------------------------------|-----------------------------|---------------------------------|--------------------|
| 1  | <i>iSDTM Version V311 Checks Summary - Run on 14JUL2011</i> |                                       |                     |                               |                             |                                 |                    |
| 2  | <i>Libname: /biostats/drug123/data</i>                      |                                       |                     |                               |                             |                                 |                    |
| 3  | <i>iSDTM Macro was run by: aboutalh on a Unix platform.</i> |                                       |                     |                               |                             |                                 |                    |
| 4  | <i>4012238 checks were performed</i>                        |                                       |                     |                               |                             |                                 |                    |
| 5  | <i>27 datasets were validated</i>                           |                                       |                     |                               |                             |                                 |                    |
| 6  |                                                             |                                       |                     |                               |                             |                                 |                    |
| 7  | Domain                                                      | Description                           | Number of Variables | Number of Character Variables | Number of Numeric Variables | Number of Physical Observations | Number of Subjects |
| 8  | AE                                                          | Adverse Events                        | 19                  | 16                            | 3                           | 10827                           | 1282               |
| 9  | CM                                                          | Concomitant Medications               | 19                  | 16                            | 3                           | 14460                           | 1290               |
| 10 | DM                                                          | Demographics                          | 17                  | 16                            | 1                           | 1430                            | 1430               |
| 11 | DS                                                          | Disposition                           | 13                  | 10                            | 3                           | 54328                           | 1430               |
| 12 | DV                                                          | Protocol Deviations                   | 7                   | 6                             | 1                           | 8318                            | 1247               |
| 13 | EG                                                          | ECG Findings                          | 18                  | 14                            | 4                           | 69205                           | 1416               |
| 14 | EX                                                          | Drug Exposure                         | 15                  | 11                            | 4                           | 1417                            | 1417               |
| 15 | HO                                                          | Hospitalization                       | 10                  | 7                             | 3                           | 350                             | 246                |
| 16 | IE                                                          | Inclusion/Exclusion Exceptions        | 10                  | 9                             | 1                           | 61                              | 51                 |
| 17 | IM                                                          | Imaging                               | 15                  | 11                            | 4                           | 9249                            | 684                |
| 18 | LB                                                          | Laboratory Test Results               | 24                  | 18                            | 6                           | 1262893                         | 1428               |
| 19 | MH                                                          | Medical History                       | 15                  | 13                            | 2                           | 50335                           | 1430               |
| 20 | PE                                                          | Physical Examinations                 | 15                  | 12                            | 3                           | 91212                           | 1430               |
| 21 | QS                                                          | Questionnaires                        | 20                  | 16                            | 4                           | 653467                          | 1430               |
| 22 | RL                                                          | Relapse                               | 13                  | 10                            | 3                           | 3630                            | 476                |
| 23 | SC                                                          | Subject Characteristics               | 12                  | 9                             | 3                           | 14349                           | 1430               |
| 24 | SU                                                          | Substance Use                         | 12                  | 10                            | 2                           | 4257                            | 1420               |
| 25 | SUPPAE                                                      | Supplemental Qualifiers for AE Domain | 10                  | 10                            | 0                           | 48755                           | 1282               |
| 26 | SUPPCM                                                      | Supplemental Qualifiers for cm Domain | 10                  | 10                            | 0                           | 16275                           | 1290               |
| 27 | SUPPDS                                                      | Supplemental Qualifiers for DS Domain | 10                  | 10                            | 0                           | 5533                            | 1430               |
| 28 | SUPPDV                                                      | Supplemental Qualifiers for dv Domain | 10                  | 10                            | 0                           | 5855                            | 974                |
| 29 | SUPPIM                                                      | Supplemental Qualifiers for im Domain | 10                  | 10                            | 0                           | 36442                           | 684                |
| 30 | SUPPLB                                                      | Supplemental Qualifiers for lb Domain | 10                  | 10                            | 0                           | 1321592                         | 1428               |
| 31 | SUPPMH                                                      | Supplemental Qualifiers for mh Domain | 10                  | 10                            | 0                           | 60243                           | 1390               |
| 32 | SUPPQS                                                      | Supplemental Qualifiers for qs Domain | 10                  | 10                            | 0                           | 128244                          | 1430               |
| 33 | VS                                                          | Vital Signs                           | 16                  | 12                            | 4                           | 137223                          | 1430               |

2. Sheet SDTM CDISC Checks:

|      | A                                                            | B                       | C             | D             | E                                                 | F                     | G                         |
|------|--------------------------------------------------------------|-------------------------|---------------|---------------|---------------------------------------------------|-----------------------|---------------------------|
| 1    | <b>iSDTM Version V311 Checks Summary - Run on 13JUL2011</b>  |                         |               |               |                                                   |                       |                           |
| 2    | <b>Libname: /biostats/drug123/data</b>                       |                         |               |               |                                                   |                       |                           |
| 3    | <b>iSDTM Macro was run by: Aboutaleb on a Unix platform.</b> |                         |               |               |                                                   |                       |                           |
| 4    | <b>4012238 checks were performed</b>                         |                         |               |               |                                                   |                       |                           |
| 5    | <b>27 datasets were validated</b>                            |                         |               |               |                                                   |                       |                           |
| 6    |                                                              |                         |               |               |                                                   |                       |                           |
| 7    | Sdtm domain Name                                             | Member Label            | Variable Name | Variable Type | Label                                             | Position              | Domain Validation Finding |
| 2274 | EG                                                           | ECG Findings            | VISITNUM      |               |                                                   | Change per SDTM guide | /Order not match          |
| 2275 | EG                                                           | ECG Findings            | VISIT         |               |                                                   | Change per SDTM guide | /Order not match          |
| 6529 | HO                                                           | Hospitalization         | HOCAT         |               | Change to 'Category for Hospitalization'          |                       | /Label not match          |
| 6530 | HO                                                           | Hospitalization         | HOSTDTC       |               | Change to 'Start Date/Time of Hospitalization'    |                       | /Label not match          |
| 6531 | HO                                                           | Hospitalization         | HOENDTC       |               | Change to 'End Date/Time of Hospitalization'      |                       | /Label not match          |
| 6532 | HO                                                           | Hospitalization         | HOSTDY        |               | Change to 'Study Day of Start of Hospitalization' |                       | /Label not match          |
| 6533 | HO                                                           | Hospitalization         | HOENDY        |               | Change to 'Study Day of End of Hospitalization'   |                       | /Label not match          |
| 6568 | LB                                                           | Laboratory Test Results | STUDYID       |               |                                                   | Change per SDTM guide | /Order not match          |
| 6569 | LB                                                           | Laboratory Test Results | DOMAIN        |               |                                                   | Change per SDTM guide | /Order not match          |
| 6570 | LB                                                           | Laboratory Test Results | USUBJID       |               |                                                   | Change per SDTM guide | /Order not match          |
| 6571 | LB                                                           | Laboratory Test Results | LBSEQ         |               |                                                   | Change per SDTM guide | /Order not match          |

**Conclusion:**

The validation of CDISC SDTM and SUPPQUAL data sets takes a lot of time and resources due to its size and complexity. However, through a tool like %iSDTM, the validation will take less time and use fewer resources, and the SAS programmer can easily validate CDISC SDTM and SUPPQUAL data sets. This tool is a low cost solution, and effective means to provide greater flexibility to users to validate the STDM domains, and follow-up on any updates to master SDTM files.

**Acknowledgment:**

The author would like to thank Ms. Tina Casteris and Mr. Vincent Da Forno for their review and valuable comments.

**TRADEMARKS:**

SAS® is a registered trademark of the SAS Institute Inc., in the USA and other countries. Windows® is a registered trademark of the Microsoft Corporation. Other brand and product names are trademarks of their respective companies.

**REFERENCES:**

- [1] *SAS® 9.2 Macro Language: Reference*. Cary, NC: SAS Institute Inc.
- [2] *SAS® 9.2 Language Reference: Concepts*. Cary, NC: SAS Institute Inc.
- [3] *SAS® 9.2 Language Reference: Dictionary*. Cary, NC: SAS Institute Inc.

**Contact Information**

Your comments and questions are valued and encouraged.

Contact the author at:

**Hany Aboutaleb**

Biogen Idec.

14 Cambridge Center

Cambridge MA 02142

Tel. #: (617) 914-7125

Fax: (617) 679-3280

Internet: [hany.aboutaleb@biogenidec.com](mailto:hany.aboutaleb@biogenidec.com)

**Sample Code used in iSD TM macro:**

```

*-----*
* Protocol:   Macro Library
* Program:    varlst.sas
*
* Programmer: Hany Aboutaleb
* Date:       10May2011
* Purpose:    to store values of variable in macro variable
* Input parameters:
*             inds: Name of the datasets that have a variable need to store in macro variable
*             byvar: Name of by variable to sort the data with
* Macro logic: PROC SQL is used to create macro variables VARlist sparated by ' '
*             (to store values of variable in macro variable)
* Platform:   HP-UX
* SAS Version: 9.1.2
*
* Modifications:
*
*-----*
%macro varlst(inds=,byvar=memname);
%global varlist;
proc sort data=&inds out=list nodupkey;
  by &byvar;
run;
proc sql noprint;
  select &byvar
  into :varlist separated by ' '
  from list;
quit;
%mend varlst;
%*-----*
* Protocol:   Macro Library
* Program:    content.sas
*
* Programmer: Hany Aboutaleb
* Date:       30May2011
* Purpose:    Project contents Code
* Input parameters:
*             inlib: Name of the libname that have data set to be processed
*             outdm: Name of library memname output content
*             outds: Name of library output data sets contents
* Platform:   HP-UX
* SAS Version: 9.1.2
*
* Modifications:
*
*-----*
%macro content(inlib=,outdm=,outds=);
proc sql;
create table &outdm as
select memname,memlabel,nobs,nvar,maxlabel,num_character,num_numeric
from dictionary.tables
where libname=%upcase("&inlib") ;
create table &outds as
select memname,name,type,length,varnum,label,format
from dictionary.columns
where libname=%upcase("&inlib") ;
quit;
%mend content;

```

```

*-----*
* Protocol:   Macro Library
* Program:    wordcount.sas
*
* Programmer: Hany Aboutaleb
* Date:       10May2011
* Purpose:    %*The %WORDCOUNT macro, which uses the %QSCAN function, can be used to
*             count the words in a list. This macro function returns the count and
*             does not create any global macro variables.
* Input parameters:
*             list: macro variables list have value of a list sparated by ' '
* Macro logic:
*             1. The count is initialized to 0.
*             2 %QSCAN looks ahead to see if the next word (&COUNT + 1) exists.
*             3 Check to see if there is a next word. If there is then enter the loop
*             so that &COUNT can be incremented 1.
*             4 Since the &COUNT+1 word was found, increment the value of &COUNT.
*             5 When the last word has been counted, exit the loop and return the
*             number of words counted by passing it back.
* Sample call:
* The following use of %WORDCOUNT displays the number of variables in &VARLIST: %wordcount(&varlist)
* Count the number of words in &LIST
* Platform:   HP-UX
* SAS Version: 9.1.2
*
* Modifications:
*-----*
%macro wordcount(list);
%local count;
%let count=0;
%do %while(%qscan(&list,&count+1,%str( )) ne %str());
%let count = %eval(&count+1);
%end;
&count
%mend wordcount;
*-----*
* Protocol:   Macro Library
* Program:    countsubj.sas
*
* Programmer: Hany Aboutaleb
* Date:       10May2011
* Purpose:    Project contents Code
* Input parameters:
*             inlib: Name of the libname that have data set to be processed
*             outcnt: Name of output file
*             ds: name of data set to get contents for (default=_all_)
* Platform:   HP-UX
* SAS Version: 9.1.2
*
* Modifications:
*-----*
%macro countsubj(inlib=,outcnt=);
%* create macro variable to store all domain *;
%varlst(inds=memlist,byvar=memname);
%let ds=;
%let sumnum=1;
%let sumds=;
%do i=1 %to %wordcount(&varlist);
%let ds=%scan(&varlist,&i);
* Using PROC SQL to count the usubjid in each domain. *;
%if %length(&ds)=2 or &ds=NORMLAB2 or %upcase(%substr(&ds,1,2))=AD %then %do;
%let dsid=%sysfunc(open(&inlib..&ds));
%let ok=1; %let ok1=1;
%let num1 = %sysfunc(varnum(&dsid,usubjid));
%if &num1=0 %then %let ok=0;

```



```

        %let num2 = %sysfunc(varnum(&dsid,subjid));
        %if &num2=0 %then %let ok1=0;
    %let rc = %sysfunc(close(&dsid));
        %put ok=&ok; %put ok1=&ok1;
        %if &ok and ^&ok1 %then %let subj=usubjid;
            %else %if ^&ok and &ok1 %then %let subj=subjid;
                %else %if &ds=DM %then %let subj=usubjid;;
%if &ok1 or &ok %then %do;
%put subj=&subj;
proc sql;
    create table d&i as
        select count(distinct(&subj)) as count,"&ds" as domain
            from &inlib..&ds;
%if &i=1 %then %do;%let sumds=d&i;%let det="&ds";%end;
    %else %do;%let sumds=&sumds d&i;%let det=&det "&ds";%end;
quit;
%end;
%end;
%put sumds=&&sumds;
%put det=&&det;
%let w = 1;
%let dsn = %scan(&sumds,&w);
data &outcnt;
length domain $6;
    set %do %while (&dsn^=);
        &dsn
        %let w = %eval(&w + 1);
        %let dsn = %scan(&sumds,&w);
    %end;;
run;
proc sort data=&outcnt;by domain;
run;
%mend countsbj;
%macro Chksdtm(stdbiib=,insdtm=);
proc sort data=&stdbiib;
    by domain variable;
run;
data chk1 chk2 chk3 chk4;
merge sdtmbiib(in=a) &insdtm(in=b);
by domain variable;
if (a and b) then output chk1;
else if (a and ^b) and core in('Req') then output chk2;
else if (a and ^b) and core in('Exp') then output chk4;
else if (^a and b) then output chk3;
run;

*** var order in cdisc;
proc sort data=chk1;
    by domain pos;
run;

data chk_1;
set chk1;
by domain pos;
retain cnt;
if first.domain then cnt=0;
cnt+1;
varnum=cnt;
run;

*** var order in crtdir mapping;
proc sort data=chk_1;
    by domain pos_;
run;

```

```

data chk1_;
  set chk_1;
  by domain pos_;
  retain cnt1;
  if first.domain then cnt1=0;
  cnt1+1;
  var_num=cnt1;
run;

proc sort data=chk1_;
  by domain;
run;

data chksdtm1(keep=domain memlabel variable chk_log type label position);
  retain domain memlabel variable chk_log type label position ;
  length domain $20 memlabel $40 variable $8 chk_log $ 500 type $ 20 label position $ 60 ;
  set chk1_(in=a rename=(domain=dm memlabel=meml)) chk2(in=b rename=(domain=dm memlabel=meml))
    chk4(in=d rename=(domain=dm memlabel=meml))  chk3(in=c rename=(domain=dm memlabel=meml));
  by dm;
  chk_log='';
  label='';
  position='';
  type='';
  domain=dm;
  memlabel=meml;
  if a then do;
    if upcase(sdtm_type) ^= upcase(type1) then do;
      chk_log=compbl(chk_log)||'/Type not match';
      type='Change to '||strip(sdtm_type);
      flag=1;
    end;

    if compbl(sdtm_label) ^= compbl(label1) then do;
      chk_log=strip(chk_log)||'/Label not match';
      label="Change to ' "||strip(sdtm_label)||"'";
      flag=1;
    end;

    if length(label1) > 40 then do;
      chk_log=strip(chk_log)||'/Label length is >40';
      label=strip(label1)||'/Revise label';
      flag=1;
    end;

    if varnum ^= var_num then do;
      chk_log=strip(chk_log)||'/Order not match';
      Position='Change per SDTM guide';
      flag=1;
    end;
  end;
  if flag=1 then output;
  if b then do;
    chk_log='SDTM required var not present: add '||compbl(variable);
    output;
  end;
  if d then do;
    chk_log='SDTM expected var not present: add '||compbl(variable);
    output;
  end;
  if c then do;
    chk_log=strip(variable)||' not in SDTM var list: drop or remove to SUPP?';
    output;
  end;
run;

data chksdtm2(keep=domain memlabel variable chk_log type label position);

```

```

retain domain memlabel variable chk_log type label position ;
length domain $20 memlabel $40 variable $8 chk_log $ 500 type $ 20 label position $ 60 ;
set member (rename=(domain=dm memlabel=meml));
by dm;
chk_log='';
  label='';
  position='';
  type='';
    variable='';
    domain=dm;
    memlabel=meml;
if memlabel='' then do;
  chk_log=strip(Domain_Label)||'Domain Data label is missing: add data label';
  label="Add Label to Domain Data: "||strip(Domain_Label)||"";
  output;
end;
if length(memlabel)>40 then do;
  chk_log='Domain Data label >40: revise Domain label';
  output;
end;
run;

data chksdtm;
set %if %sysfunc(exist(chksdtm1))^=0 %then chksdtm1;%if %sysfunc(exist(chksdtm2))^=0 %then chksdtm2;;
run;

%let obs=.;
data _null_;
  set chksdtm end=eof;
  n+1;
  if eof then call symputx('obs', n);
run;
%put <===== obs = &obs =====>;

%if &obs=. %then %do;
  data chksdtm;
    length domain $20 memlabel $40 variable $8 chk_log $ 500 type $ 20 label position $ 60 ;
    chk_log="Passed SDTM check!";
    output;
  run;
%end;
%mend chksdtm;
*****
*
* Program Name:  iSDTM
*
* Developed By:  Hany Aboutaleb (Cambridge, x(4)7125)s
*
* Date:         10/1/10
*
* Purpose:      Macro to Create Excel Work sheet
* Input parameters:
*               inlib: Name of the libname that have data set to be processed
*               excelout: Name of the Excel file output (default=isdtdm.xls)
*               sdtm: Which version of SDTM want the macro to use (default=V311)
*               chk_req: Verifies that all required variable fields do not contain missing values (default=Y)
*               chk_exp: Detects occurrences of expected variable fields that contain missing values (default=Y)
*               chk_iso: Detects the conformance of all ISO 8601 assigned values, including date, time, datetime,
*                       duration, and interval types (default=Y)
*               Debug: to debug the macro (YES/NO) (default: NO)
*               cleanup: clean up temporarily data sets(YES/NO)(default=YES)
*               Version: Version control to the macro for future use in case of new
*                       release to the macro (default=1)
* Output file:  isdtdm.xls
*
* Modifications:
*****;

%macro isdtdm(inlib=, excelout=isdtdm,sdtm=v311,debug=N,chk_req=Y,chk_exp=Y,chk_iso=Y,version=1,cleanup=YES);
  %put;

```

```

%put =====>>>;
%put <> Biogenidec System Macro iSDTM:Macro to create excel worksheet for you SAS with SDTM data ;
%put =====>>>;

%if %upcase(%substr(&debug,1,1)) = Y %then
  %do;
    options symbolgen mlogic mprint;
    %put _user_;
  %end;

%local er ror war ning keepvars;
%let parmerr=0;
%let er = ER ;
%let ror = ROR ;
%let war = WAR ;
%let ning=NING ;
%let keepvars=domain variable varnum label len action biibaction;
  %LET PAGEOF=Page ^{thispage} of ^{lastpage} ;

  ODS escapechar='^' ;

%include "/biostats/macros/dev/DataQuality/mac/macdataq.sas";

%*-----
  Validate parameters
  -----;

** Break if parameters libin or dm empty ;
    %if ^(%qcmpres(&inlib) eq ) %then
      %do;
        %if ( %sysfunc(libref( &inlib )) ne 0 ) %then
          %do;
            %put %sysfunc(sysmsg());
            %put &er&ror: Input library %str(&inlib) is not correctly defined.;
            %let parmerr=1;
          %end;
        %end;
      %end;

%*-----
  Quit the macro program if you have any wrong input parameters
  -----;
%if (&parmerr) %then %return;

%*-----;

%if &version=1 %then
  %do;
%*chk_sdtm data result;
%sdm(inlib=&inlib,sdtm=&sdtm,chk_req=&chk_req,chk_exp=&chk_exp,chk_iso=&chk_iso,chk_sdtm=y,chk_supp=y);
%*summary data result;
%summary(inlib=&inlib);
data summary;
set summary;
by domain;
retain i 0 ;
if first.domain then i = i + 1 ;
  if mod(i,2) = 0 then blue = "Y" ;
  else blue = "N" ;
run ;

data chk_sdtm;
set chk_sdtm;
by domain;
retain sortvar 0;
if first.domain then sortvar = sortvar + 1 ;
i=_n_;
  if mod(i,2) = 0 then green = "Y" ;
  else green = "N" ;
run ;

%if (&sysscp = WIN) %then %do ;
%let chrOSName = Windows ;
%end ;
  %else %do ;
    %let chrOSName = Unix ;
  %end ;
%*=====

```

```

Generate the Excel spreadsheet
=====;
options nodate nonumber nobyline missing= ' ' ;
ods listing close;
ods tagsets.ExcelXP file="%q%lowercase(&excelout).xls" style=sansPrinter ;
ods tagsets.ExcelXP options( autofilter='all'
                             orientation="landscape"
                             embedded_titles='yes'
                             PAGES_FITWIDTH="2"
                             gridlines="yes"
                             blackandwhite="yes"
                             zoom='85'
                             scale='75'
                             autofit_height='yes'
                             width_fudge='0.75'
                             print_footer='&C Page &P of &N'
                             );
ods tagsets.ExcelXP options( frozen_rowheaders='yes'
                             ABSOLUTE_COLUMN_WIDTH='8, 25, 10, 10, 10, 10,10,10'
                             sheet_name='SDTM Summary Checks'
                             row_repeat='1-8'
                             autofit_height='yes'
                             width_fudge='0.75'
                             );
title1 "iSDTM Version &sdm Checks Summary - Run on &sysdate9 ";
title2 "Libname: %sysfunc(pathname(&inlib.))";
title3 "iSDTM Macro was run by: &sysuserid on a &chrOSName platform.";
title4 "&totobs checks were performed";
title5 "&ndom datasets were validated";
footnote1 j=1 "Output: &excelout.xls" j=c "Created on &sysdate &system" j=r "&PAGEOF";

proc report data=SUMMARY MISSING ;
col i blue domain memlabel nvar num_character num_numeric nobis count;
define i /order order=data noprint ;
define blue /display noprint ;
COMPUTE blue ;
if (blue = "Y") then call define(_row_, 'style', 'style=[background=#66ccff]') ;
ENDCOMP ;
run;
ods tagsets.ExcelXP options( frozen_rowheaders='yes'
                             ABSOLUTE_COLUMN_WIDTH='8, 14, 10, 12,12,12, 25'
                             sheet_name='SDTM CDISC Checks'
                             row_repeat='1-5'
                             autofit_height='yes'
                             width_fudge='0.75'
                             );
title1 "iSDTM Version &sdm Checks Summary - Run on &sysdate9 ";
title2 "Libname: %sysfunc(pathname(&inlib.))";
title3 "iSDTM Macro was run by: &sysuserid on a &chrOSName platform.";
title4 "&totobs checks were performed";
title5 "&ndom datasets were validated";
footnote1 j=1 "Output: &excelout.xls" j=c "Created on &sysdate &system" j=r "&PAGEOF";
proc report data=chk_sdtm MISSING ;
col sortvar green domain memlabel variable type label position chk_log;
define sortvar /order order=data noprint ;
define green /display noprint ;
break after sortvar / skip;
COMPUTE green ;
if (green = "Y") then call define(_row_, 'style', 'style=[background=#99ffff]') ;
ENDCOMP ;
run;
%if %upcase(%substr(&cleanup,1,1)) = Y %then
%DO;
proc datasets library=work memtype=data nolist nowarn;
save chk_sdtm SUMMARY;
Quit;
run;
%end;

%end; * version do loop;
%if %upcase(%substr(&debug,1,1)) = Y %then
%do;
%put _user_;
%end;
%mend isdtm;

```